

NAG Library Routine Document

D02BJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

D02BJF integrates a system of first-order ordinary differential equations over an interval with suitable initial conditions, using a fixed order Runge–Kutta method, until a user-specified function, if supplied, of the solution is zero, and returns the solution at points specified by you, if desired.

2 Specification

```
SUBROUTINE D02BJF (X, XEND, N, Y, FCN, TOL, RELABS, OUTPUT, G, W, IFAIL)
  INTEGER          N, IFAIL
  REAL (KIND=nag_wp) X, XEND, Y(N), TOL, G, W(20*N)
  CHARACTER(1)    RELABS
  EXTERNAL        FCN, OUTPUT, G
```

3 Description

D02BJF advances the solution of a system of ordinary differential equations

$$y'_i = f_i(x, y_1, y_2, \dots, y_n), \quad i = 1, 2, \dots, n,$$

from $x = X$ to $x = XEND$ using a fixed order Runge–Kutta method. The system is defined by FCN, which evaluates f_i in terms of x and $y = (y_1, y_2, \dots, y_n)$. The initial values of $y = (y_1, y_2, \dots, y_n)$ must be given at $x = X$.

The solution is returned via the OUTPUT supplied by you and at points specified by you, if desired: this solution is obtained by C^1 interpolation on solution values produced by the method. As the integration proceeds a check can be made on the user-specified function $g(x, y)$ to determine an interval where it changes sign. The position of this sign change is then determined accurately by C^1 interpolation to the solution. It is assumed that $g(x, y)$ is a continuous function of the variables, so that a solution of $g(x, y) = 0$ can be determined by searching for a change in sign in $g(x, y)$. The accuracy of the integration, the interpolation and, indirectly, of the determination of the position where $g(x, y) = 0$, is controlled by the arguments TOL and RELABS.

4 References

Shampine L F (1994) *Numerical solution of ordinary differential equations* Chapman and Hall

5 Arguments

1: X – REAL (KIND=nag_wp) *Input/Output*

On entry: the initial value of the independent variable x .

On exit: if g is supplied by you, it contains the point where $g(x, y) = 0$, unless $g(x, y) \neq 0$ anywhere on the range X to XEND, in which case, X will contain XEND (and the error indicator IFAIL = 6 is set); if g is not supplied by you it contains XEND. However, if an error has occurred, it contains the value of x at which the error occurred.

- 2: XEND – REAL (KIND=nag_wp) *Input*
On entry: the final value of the independent variable. If XEND < X, integration will proceed in the negative direction.
Constraint: XEND \neq X.
- 3: N – INTEGER *Input*
On entry: n , the number of equations.
Constraint: N > 0.
- 4: Y(N) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the initial values of the solution y_1, y_2, \dots, y_n at $x = X$.
On exit: the computed values of the solution at the final point $x = X$.
- 5: FCN – SUBROUTINE, supplied by the user. *External Procedure*
FCN must evaluate the functions f_i (i.e., the derivatives y'_i) for given values of its arguments x, y_1, \dots, y_n .

The specification of FCN is:

```
SUBROUTINE FCN (X, Y, F)
```

```
REAL (KIND=nag_wp) X, Y(*), F(*)
```

1: X – REAL (KIND=nag_wp) *Input*

On entry: x , the value of the independent variable.

2: Y(*) – REAL (KIND=nag_wp) array *Input*

On entry: y_i , for $i = 1, 2, \dots, n$, the value of the variable.

3: F(*) – REAL (KIND=nag_wp) array *Output*

On exit: the value of f_i , for $i = 1, 2, \dots, n$.

FCN must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub) program from which D02BJF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 6: TOL – REAL (KIND=nag_wp) *Input*
On entry: a **positive** tolerance for controlling the error in the integration. Hence TOL affects the determination of the position where $g(x, y) = 0$, if g is supplied.

D02BJF has been designed so that, for most problems, a reduction in TOL leads to an approximately proportional reduction in the error in the solution. However, the actual relation between TOL and the accuracy achieved cannot be guaranteed. You are strongly recommended to call D02BJF with more than one value for TOL and to compare the results obtained to estimate their accuracy. In the absence of any prior knowledge, you might compare the results obtained by calling D02BJF with RELABS = 'D' and with each of TOL = 10.0^{-p} and TOL = 10.0^{-p-1} where p correct significant digits are required in the solution, y . The accuracy of the value x such that $g(x, y) = 0$ is indirectly controlled by varying TOL. You should experiment to determine this accuracy.

Constraint: $10.0 \times \text{machine precision} < \text{TOL} < 0.01$.

7: RELABS – CHARACTER(1)

Input

On entry: the type of error control. At each step in the numerical solution an estimate of the local error, *est*, is made. For the current step to be accepted the following condition must be satisfied:

$$est = \max(e_i / (\tau_r \times \max(|y_i|, \tau_a))) \leq 1.0$$

where τ_r and τ_a are defined by

RELABS	τ_r	τ_a
'M'	TOL	1.0
'A'	ϵ_r	TOL/ ϵ_r
'R'	TOL	ϵ_a
'D'	TOL	ϵ_a

where ϵ_r and ϵ_a are small machine-dependent numbers and e_i is an estimate of the local error at y_i , computed internally. If the condition is not satisfied, the step size is reduced and the solution is recomputed on the current step. If you wish to measure the error in the computed solution in terms of the number of correct decimal places, then RELABS should be set to 'A' on entry, whereas if the error requirement is in terms of the number of correct significant digits, then RELABS should be set to 'R'. If you prefer a mixed error test, then RELABS should be set to 'M', otherwise if you have no preference, RELABS should be set to the default 'D'. Note that in this case 'D' is taken to be 'R'.

Constraint: RELABS = 'M', 'A', 'R' or 'D'.

8: OUTPUT – SUBROUTINE, supplied by the NAG Library or the user. *External Procedure*

OUTPUT permits access to intermediate values of the computed solution (for example to print or plot them), at successive user-specified points. It is initially called by D02BJF with XSOL = X (the initial value of x). You must reset XSOL to the next point (between the current XSOL and XEND) where OUTPUT is to be called, and so on at each call to OUTPUT. If, after a call to OUTPUT, the reset point XSOL is beyond XEND, D02BJF will integrate to XEND with no further calls to OUTPUT; if a call to OUTPUT is required at the point XSOL = XEND, then XSOL must be given precisely the value XEND.

The specification of OUTPUT is:

```
SUBROUTINE OUTPUT (XSOL, Y)
```

```
REAL (KIND=nag_wp) XSOL, Y(*)
```

```
1: XSOL – REAL (KIND=nag_wp)
```

Input/Output

On entry: the output value of the independent variable x .

On exit: you must set XSOL to the next value of x at which OUTPUT is to be called.

```
2: Y(*) – REAL (KIND=nag_wp) array
```

Input

On entry: the computed solution at the point XSOL.

OUTPUT must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which D02BJF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

If you do not wish to access intermediate output, the actual argument OUTPUT **must** be the dummy routine D02BJX. (D02BJX is included in the NAG Library.)

9: G – REAL (KIND=nag_wp) FUNCTION, supplied by the user.

External Procedure

G must evaluate the function $g(x, y)$ for specified values x, y . It specifies the function g for which the first position x where $g(x, y) = 0$ is to be found.

The specification of G is:

```
FUNCTION G (X, Y)
REAL (KIND=nag_wp) G
REAL (KIND=nag_wp) X, Y(*)
```

where n is the value of N in the call of D02BJF.

1:	X – REAL (KIND=nag_wp)	<i>Input</i>
	<i>On entry:</i> x , the value of the independent variable.	
2:	Y(*) – REAL (KIND=nag_wp) array	<i>Input</i>
	<i>On entry:</i> y_i , for $i = 1, 2, \dots, n$, the value of the variable.	

G must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub) program from which D02BJF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

If you do not require the root-finding option, the actual argument G **must** be the dummy routine D02BJW. (D02BJW is included in the NAG Library.)

10:	W(20 × N) – REAL (KIND=nag_wp) array	<i>Workspace</i>
11:	IFAIL – INTEGER	<i>Input/Output</i>

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, TOL \geq 0.01,
or TOL is too small
or $N \leq 0$,
or RELABS \neq 'M', 'A', 'R' or 'D',
or X = XEND.

IFAIL = 2

With the given value of TOL, no further progress can be made across the integration range from the current point $x = X$. (See Section 9 for a discussion of this error exit.) The components Y(1), Y(2), ..., Y(N) contain the computed values of the solution at the current point $x = X$. If you have supplied g , then no point at which $g(x, y)$ changes sign has been located up to the point $x = X$.

IFAIL = 3

TOL is too small for D02BJF to take an initial step. X and Y(1), Y(2), ..., Y(N) retain their initial values.

IFAIL = 4

XSOL has not been reset or XSOL lies behind X in the direction of integration, after the initial call to OUTPUT, if the OUTPUT option was selected.

IFAIL = 5

A value of XSOL returned by the OUTPUT has not been reset or lies behind the last value of XSOL in the direction of integration, if the OUTPUT option was selected.

IFAIL = 6

At no point in the range X to XEND did the function $g(x, y)$ change sign, if g was supplied. It is assumed that $g(x, y) = 0$ has no solution.

IFAIL = 7

A serious error has occurred in an internal call to an interpolation routine. Check all (sub) program calls and array dimensions. Seek expert help.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The accuracy of the computation of the solution vector Y may be controlled by varying the local error tolerance TOL. In general, a decrease in local error tolerance should lead to an increase in accuracy. You are advised to choose RELABS = 'D' unless you have a good reason for a different choice.

If the problem is a root-finding one, then the accuracy of the root determined will depend on the properties of $g(x, y)$ and on the values of TOL and RELABS. You should try to code G without introducing any unnecessary cancellation errors.

8 Parallelism and Performance

D02BJF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

If more than one root is required, then to determine the second and later roots D02BJF may be called again starting a short distance past the previously determined roots. Alternatively you may construct your own root-finding code using C05AZF, D02PFF and D02PSF.

If D02BJF fails with `IFAIL = 3`, then it can be called again with a larger value of `TOL` if this has not already been tried. If the accuracy requested is really needed and cannot be obtained with this routine, the system may be very stiff (see below) or so badly scaled that it cannot be solved to the required accuracy.

If D02BJF fails with `IFAIL = 2`, it is probable that it has been called with a value of `TOL` which is so small that a solution cannot be obtained on the range `X` to `XEND`. This can happen for well-behaved systems and very small values of `TOL`. You should, however, consider whether there is a more fundamental difficulty. For example:

- (a) in the region of a singularity (infinite value) of the solution, the routine will usually stop with `IFAIL = 2`, unless overflow occurs first. Numerical integration cannot be continued through a singularity, and analytic treatment should be considered;
- (b) for ‘stiff’ equations where the solution contains rapidly decaying components, the routine will use very small steps in x (internally to D02BJF) to preserve stability. This will exhibit itself by making the computing time excessively long, or occasionally by an exit with `IFAIL = 2`. Runge–Kutta methods are not efficient in such cases, and you should try D02EJF.

10 Example

This example illustrates the solution of four different problems. In each case the differential system (for a projectile) is

$$\begin{aligned}y' &= \tan \phi \\v' &= \frac{-0.032 \tan \phi}{v} - \frac{0.02v}{\cos \phi} \\ \phi' &= \frac{-0.032}{v^2}\end{aligned}$$

over an interval `X = 0.0` to `XEND = 10.0` starting with values $y = 0.5$, $v = 0.5$ and $\phi = \pi/5$. We solve each of the following problems with local error tolerances $1.0\text{E}-4$ and $1.0\text{E}-5$.

- (i) To integrate to $x = 10.0$ producing intermediate output at intervals of 2.0 until a root is encountered where $y = 0.0$.
- (ii) As (i) but with no intermediate output.
- (iii) As (i) but with no termination on a root-finding condition.
- (iv) As (i) but with no intermediate output and no root-finding termination condition.

10.1 Program Text

```
! D02BJF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d02bjfe_mod

! Data for D02BJF example program

! .. Use Statements ..
Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
Implicit None
! .. Accessibility Statements ..
Private
Public :: fcn, g, output
```

```

! .. Parameters ..
Integer, Parameter, Public      :: n = 3, nin = 5, nout = 6
! .. Local Scalars ..
Real (Kind=nag_wp), Public, Save :: h, xend
! n: number of differential equations
Contains
Subroutine output(xsol,y)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: xsol
! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: y(*)
! .. Local Scalars ..
Integer      :: j
! .. Intrinsic Procedures ..
Intrinsic    :: abs
! .. Executable Statements ..
Write (nout,99999) xsol, (y(j),j=1,n)
xsol = xsol + h
! Make sure we exactly hit xsol = xend
If (abs(xsol-xend)<h/4.0E0_nag_wp) Then
  xsol = xend
End If
Return

99999  Format (1X,F8.2,3F13.4)
End Subroutine output
Subroutine fcn(x,y,f)

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: alpha = -0.032E0_nag_wp
Real (Kind=nag_wp), Parameter :: beta = -0.02E0_nag_wp
! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: x
! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: f(*)
Real (Kind=nag_wp), Intent (In) :: y(*)
! .. Intrinsic Procedures ..
Intrinsic    :: cos, tan
! .. Executable Statements ..
f(1) = tan(y(3))
f(2) = alpha*tan(y(3))/y(2) + beta*y(2)/cos(y(3))
f(3) = alpha/y(2)**2
Return
End Subroutine fcn
Function g(x,y)

! .. Function Return Value ..
Real (Kind=nag_wp)      :: g
! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: x
! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: y(*)
! .. Executable Statements ..
g = y(1)
Return
End Function g
End Module d02bjfe_mod
Program d02bjfe

! D02BJF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d02bjf, d02bjw, d02bjx, nag_wp
Use d02bjfe_mod, Only: fcn, g, h, n, nin, nout, output, xend
! .. Implicit None Statement ..
Implicit None
! .. Local Scalars ..
Real (Kind=nag_wp)      :: tol, x, xinit
Integer                 :: i, icase, ifail, iw, j, kinit
! .. Local Arrays ..

```

```

      Real (Kind=nag_wp), Allocatable :: w(:), y(:), yinit(:)
! .. Intrinsic Procedures ..
      Intrinsic :: real
! .. Executable Statements ..
      Write (nout,*) 'D02BJF Example Program Results'
      iw = 20*n
      Allocate (w(iw),y(n),yinit(n))
! Skip heading in data file
      Read (nin,*)
! xinit: initial x value, xend: final x value.
! yinit: initial solution values
      Read (nin,*) xinit, xend
      Read (nin,*) yinit(1:n)
      Read (nin,*) kinit
      Do ica = 1, 4
        Write (nout,*)
        Select Case (ica)
          Case (1)
            Write (nout,99995) ica, 'intermediate output, root-finding'
          Case (2)
            Write (nout,99995) ica, 'no intermediate output, root-finding'
          Case (3)
            Write (nout,99995) ica, 'intermediate output, no root-finding'
          Case (4)
            Write (nout,99995) ica,
              'no intermediate output, no root-finding ( integrate to XEND)'
        End Select
        Do j = 4, 5
          tol = 10.0E0_nag_wp**(-j)
          Write (nout,*)
          Write (nout,99999) ' Calculation with TOL =', tol
          x = xinit
          y(1:n) = yinit(1:n)
          If (ica/=2) Then
            Write (nout,*) ' X Y(1) Y(2) Y(3)'
            h = (xend-x)/real(kinit+1,kind=nag_wp)
          End If
! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
          ifail = 0
          Select Case (ica)
            Case (1)
              Call d02bjf(x,xend,n,y,fcn,tol,'Default',output,g,w,ifail)
              Write (nout,99998) ' Root of Y(1) = 0.0 at', x
              Write (nout,99997) ' Solution is', (y(i),i=1,n)
            Case (2)
              Call d02bjf(x,xend,n,y,fcn,tol,'Default',d02bjx,g,w,ifail)
              Write (nout,99998) ' Root of Y(1) = 0.0 at', x
              Write (nout,99997) ' Solution is', (y(i),i=1,n)
            Case (3)
              Call d02bjf(x,xend,n,y,fcn,tol,'Default',output,d02bjw,w,ifail)
            Case (4)
              Write (nout,99996) x, (y(i),i=1,n)
              Call d02bjf(x,xend,n,y,fcn,tol,'Default',d02bjx,d02bjw,w,ifail)
              Write (nout,99996) x, (y(i),i=1,n)
          End Select
        End Do
        If (ica<4) Then
          Write (nout,*)
        End If
      End Do
99999 Format (1X,A,E8.1)
99998 Format (1X,A,F7.3)
99997 Format (1X,A,3F13.4)
99996 Format (1X,F8.2,3F13.4)
99995 Format (1X,'Case ',I1,': ',A)
      End Program d02bjfe

```

10.2 Program Data

D02BJF Example Program Data

```

0.0 10.0                : xinit
0.5 0.5 6.28318530717958647692E-1 : yinit
4                                : kinit

```

10.3 Program Results

D02BJF Example Program Results

Case 1: intermediate output, root-finding

Calculation with TOL = 0.1E-03

X	Y(1)	Y(2)	Y(3)
0.00	0.5000	0.5000	0.6283
2.00	1.5493	0.4055	0.3066
4.00	1.7423	0.3743	-0.1289
6.00	1.0055	0.4173	-0.5507

Root of Y(1) = 0.0 at 7.288
Solution is -0.0000 0.4749 -0.7601

Calculation with TOL = 0.1E-04

X	Y(1)	Y(2)	Y(3)
0.00	0.5000	0.5000	0.6283
2.00	1.5493	0.4055	0.3066
4.00	1.7423	0.3743	-0.1289
6.00	1.0055	0.4173	-0.5507

Root of Y(1) = 0.0 at 7.288
Solution is 0.0000 0.4749 -0.7601

Case 2: no intermediate output, root-finding

Calculation with TOL = 0.1E-03

Root of Y(1) = 0.0 at 7.288
Solution is -0.0000 0.4749 -0.7601

Calculation with TOL = 0.1E-04

Root of Y(1) = 0.0 at 7.288
Solution is 0.0000 0.4749 -0.7601

Case 3: intermediate output, no root-finding

Calculation with TOL = 0.1E-03

X	Y(1)	Y(2)	Y(3)
0.00	0.5000	0.5000	0.6283
2.00	1.5493	0.4055	0.3066
4.00	1.7423	0.3743	-0.1289
6.00	1.0055	0.4173	-0.5507
8.00	-0.7460	0.5130	-0.8537
10.00	-3.6283	0.6333	-1.0515

Calculation with TOL = 0.1E-04

X	Y(1)	Y(2)	Y(3)
0.00	0.5000	0.5000	0.6283
2.00	1.5493	0.4055	0.3066
4.00	1.7423	0.3743	-0.1289
6.00	1.0055	0.4173	-0.5507
8.00	-0.7459	0.5130	-0.8537
10.00	-3.6282	0.6333	-1.0515

Case 4: no intermediate output, no root-finding (integrate to XEND)

Calculation with TOL = 0.1E-03

X	Y(1)	Y(2)	Y(3)
0.00	0.5000	0.5000	0.6283
10.00	-3.6283	0.6333	-1.0515

Calculation with TOL = 0.1E-04

X	Y(1)	Y(2)	Y(3)
0.00	0.5000	0.5000	0.6283
10.00	-3.6282	0.6333	-1.0515

